Small investigation into problem 11

Nasser Abbasi, 9/26/07

The problem :
Let U_1, U_2, U_3 be independent random variables uniform on [0,1]. Find the probability of the roots of the quadratic $U_1 x^2 + U_2 x + U_3$ are real
Answer:
Roots are real when discriminant is ≥ 0

In[25]:=
```
eq = U1 x^2 + U2 x + U3;
expr = First@Solve[eq == 0, x];
f = First@Cases[expr, Sqrt[any_] → any, Infinity] (*Pull out the expression under the sqrt *)
```

Out[27]=
$U2^2 - 4\, U1\, U3$

Hence we want to find $P(U2^2 - 4\,U1\,U3 > 0)$

This is the VOLUME between the above surface and between a cube of side 1. i.e. a cube of volume 1.

As an initial look, One way to view this is to look at the constant surface contours in 3 D space. We can look at the constant contour surfaces in which the function $U2^2 - 4\,U1\,U3$ is zero. And then look at the surface in which this function is on the positive side and then on the negative side of it. This will give us an idea where the volume of interest lies in relation to the zero contour surface.

In[28]:=
```
ContourPlot3D[f , {U1, 0, 1}, {U2, 0, 1}, {U3, 0, 1}, AxesLabel → {"u1", "u2", "u3"},
 Contours → {0, -.5, .5}, ContourStyle → {Yellow, Red, Green}]
```

Out[28]=



We see from the above that surfaces below the yellow surface (the GREEN) are positive, and those above it (RED) are negative. We can get a better view of the volume by getting a plot of the region where such a function is POSITIVE. Next we draw the solid region where this function is POSITIVE

In[29]:=
```
RegionPlot3D[f > 0, {U1, 0, 1}, {U2, 0, 1}, {U3, 0, 1},
 AxesLabel → {"u1", "u2", "u3"}, PlotStyle → Directive[Yellow, Opacity[0.5]], Mesh → None]
```

Out[29]=



In[30]:=
```
RegionPlot3D[f > 0, {U1, 0, 1}, {U2, 0, 1}, {U3, 0, 1},
 AxesLabel → {"u1", "u2", "u3"}, PlotStyle → Directive[Yellow], Mesh → None]
```

Out[30]=



So the solid volume in the above represents the numerical value of the probability we are looking for. It is hard for me now to see the regions of integration analytically, but there is a simulation run which gives an approximate value for the probability we need

In[31]:=
```
n = 3 000 000; SeedRandom[010 101];
u1 = Table[RandomReal[{0, 1}], {i, n}];
u2 = Table[RandomReal[{0, 1}], {i, n}];
u3 = Table[RandomReal[{0, 1}], {i, n}];
r = Select[u2^2 - 4 u1 u3, # ≥ 0 &];
Print["Probability is ", N[Length[r] / n]]
```

```
Probability is 0.253976
```